

Guia de introdução às funções do LibreOffice Calc

Direitos autorais

Copyright © 2013 Gustavo Buzzatti Pacheco

Você está autorizado a distribuir e/ou copiar este documento sob os termos da licença GNU General Public License (http://www.gnu.org/licenses/gpl.html), versão 3 ou posterior, os sob os termos da licença Creative Commons Attribution License (http://creativecommons.org/licenses/by/3.0/), versão 3.0 ou posterior.

Todas as marcas registradas neste documento pertencem aos seus respectivos proprietários.

Colaboradores

A versão original deste capítulo teve como autor:

Gustavo Buzzatti Pacheco – *gbpacheco@opendocument.com.br*

A versão original deste capítulo teve como revisora:

Vera Lucia Cavalcante Pereira

Comentários e sugestões

Qualquer comentário ou sugestão sobre esse documento pode ser enviado para o autor ou para a lista de documentação do projeto LibreOffice: docs@pt-br.libreoffice.org.

Data de publicação e versão do software

Publicado em 10 de janeiro de 2013. Baseado no LibreOffice 3.6.

Nota para usuários de Mac

Algumas teclas de atalho e itens de menu são diferentes, na plataforma Mac, daquelas utilizadas nas plataformas Windows e Linux. A tabela abaixo apresenta uma lista comparativa de alguns caminhos e teclas utilizados neste capítulo. Para uma lista completa, veja a ajuda do LibreOffice.

Windows/Linux	Equivalente no Mac	Resultado
Seleção do menu Ferramentas → Opções	LibreOffice → Preferências	Acessa as opções de configuração
Clique com botão direito	Control+clique	Abre o menu de contexto
Ctrl (Control)	業 (Comando)	Utilizada com outras teclas
F5	Shift+第+F5	Abre o Navegador
F11	∺ + <i>T</i>	Abre a janela <i>Estilos</i> e formatação

Conteúdo

Direitos autorais	2
Nota para usuários de Mac	2
Introdução	5
Primeiros passos no Calc	6
Editando o conteúdo de uma célula	
Assistente de funções	6
Operadores	8
Realce de valor	8
Séries de preenchimento	11
Fixação de referências de célula	11
Detetive	12
Atingir meta	13
Funções de Banco de Dados	15
BDSOMA	15
BDCONTAR	16
Funções de Data e Hora	17
ANO	17
AGORA	17
DIA	17
DIATRABALHOTOTAL	17
DOMINGODEPÁSCOA	18
ÉANOBISSEXTO	
FIMMÊS	
HOJE	
MÊS	
Funções Estatísticas	
DESVPAD	20
MAIOR	
MÁXIMO	
MÉDIA	
MENOR	
MÍNIMO	
Funções de Informações	
ÉERROS	
É.NÃO.DISP	24
Funções Lógicas	
SE	26
Funções Matemáticas	27
ABS	
ALEATÓRIO	
ARRED	27

ARREDONDAR.PARA.BAIXO	27
ARREDONDAR.PARA.CIMA	28
CONT.NÚM	29
CONT.SE	30
CONT.VALORES	31
CONTAR.VAZIO	32
SINAL	33
SOMASE	34
SUBTOTAL	35
Funções de Planilha	37
CORRESP	37
DESLOC	39
ESTILO	40
ÍNDICE	42
LINHA	44
PROC	44
PROCV	45
Funções de Texto	46
CARACT	46
CONCATENAR	47
DIREITA	47
ESQUERDA	48
EXT.TEXTO	48
TEXTO	48
VALOR	48
MAIÚSCULA	48
NÚM.CARACT	49
PRI.MAIÚSCULA	49
Funções Matriciais	50
TRANSPOR	50
SOMARPRODUTO	51
Funções de Suplemento	53
CONVERTER_ADD	
SE + É.NÃO.DISP + PROCV	
SOMA e SE em fórmulas matriciais	
Índice alfabético de funcões	

Introdução

O domínio dos conceitos que envolvem o uso de fórmulas e funções são fundamentais para a obtenção de bons resultados em qualquer aplicação de planilha eletrônica. Este guia tem como objetivo apresentar uma introdução concisa ao usuário que deseja aprofundar seus conhecimentos no tema, com base nas funções disponíveis no LibreOffice Calc. Nosso objetivo não é detalhar todas as funções do Calc, mas sim, apresentar com qualidade as funções mais comuns e úteis aos usuários.

O documento é dividido em duas partes. A primeira relaciona recursos úteis do Calc para o usuário que trabalha com fórmulas e funções. A segunda apresenta o detalhamento das funções, com exemplos que simulam situações reais de uso de fórmulas em planilhas, muitas vezes relacionando mais de uma função em um mesmo cenário.

As funções estão listadas por categoria, ou seja, estão organizadas da mesma forma como podem ser encontradas no assistente de funções do Calc. Para uma pesquisa alfabética, entretanto, foi disponibilizado um índice alfabético de funções, na última página (pág. 61).

Este guia estará em constante atualização com base no versionamento estável do LibreOffice para uso corporativo (versões a partir de x.x.4). Correções, comentários e contribuições de exemplos de construção de fórmulas são bem-vindas e podem ser feitas diretamente para o autor original do documento ou para a lista de documentação do projeto LibreOffice, cujos endereços estão disponíveis na seção Comentários e sugestões na página 2.

Introdução 5

Primeiros passos no Calc

Editando o conteúdo de uma célula

O primeiro passo para editar um conteúdo de célula é fazer com que seja colocada no modo de edição. Como você já deve ter percebido, o comportamento natural do cursor na planilha está definido para movimentação. No momento em que desejamos editar um conteúdo, devemos indicar para a planilha que ela deverá colocar a célula selecionada em modo de edição.

Para isso, podemos utilizar os seguintes caminhos:


- simplesmente digitar o texto desejado numa digitação normal, a planilha identificará que a digitação de um texto qualquer indica um novo conteúdo de célula;
- clicando duas vezes com o cursor do mouse sobre a célula nesse caso, a célula passará a ser editada mantendo o conteúdo anterior disponível, caso o mesmo exista;
- teclando a tecla de função F2 a tecla F2 também abre a célula corrente para edição;
- clicando sobre a linha de entrada a linha de entrada, além de exibir o conteúdo de uma célula, serve, também, para modificação.



Na figura acima, o usuário está editando a célula B2 e digitando o conteúdo na linha de entrada. Note que, ao iniciar a edição, a barra de fórmulas foi alterada, passando a incluir os botões de Cancelar (X) e Aceitar (V). Ao final de uma edição deveremos confirmar ou cancelar o conteúdo editado. Para aceitar, tecle em Enter ou no botão Aceitar (V). Para Cancelar, tecle em Esc ou clique no botão Cancelar (X).


Assistente de funções

Funções são procedimentos baseados em operações e operandos que, manipulados, retornam um determinado resultado. Funções podem simplesmente representar a implementação de operadores, como a função Soma, que veremos adiante, ou, de forma mais complexa, realizar cálculos de nível avançado.




Para inserir uma função através do assistente, siga os seguintes passos:

- 1. selecione a célula onde será inserida a função;
- 2. selecione uma das opções abaixo:
- (a) vá até o menu Inserir > Função ou
- (b) teclar Ctrl + F2 ou
- (c) clique sobre o botão **Assistente de funções**, na Barra de fórmulas.
 - 3. será aberta a tela do Assistente de funções. Selecione, então, uma categoria de função na caixa **Categoria**;



- 4. selecione o nome da função e clique no botão **Próximo >>**;
- 5. preencha os argumentos solicitados para a função;



6. clique **OK**. A fórmula será inserida na célula e o resultado será contabilizado.

Operadores

As tabelas abaixo apresentam os símbolos de operadores utilizados pelo LibreOffice Calc. Os operadores podem ser utilizados em fórmulas independentemente do uso de funções.

Operadores de referência

Sinal	Operação
:	intervalo
;	união

Operadores aritméticos

Sinal	Operação
+	adição
-	subtração
*	multiplicação
/	divisão
%	percentual
۸	exponencial

Operadores comparativos

Sinal	Operador
=	igual
>	maior
<	menor
>=	maior igual
<=	menor igual
<>	diferente


Operadores de texto

Sinal	Operador
&	concatenação

Realce de valor

A função Realce de valor destaca os conteúdos das células através da diferenciação da cor da fonte. Para acioná-la, vá até o menu **Exibir > Realce de valor** ou clique na combinação de teclas **Ctrl+F8**.

Textos são apresentados em preto, números em azul e fórmulas em verde. Essa configuração de cores é padrão para qualquer instalação do LibreOffice. Na figura abaixo, é possível identificar as três categorias. No exemplo, o conteúdo da célula B30 é a fórmula =1+1.



Note que a célula B32, que contém uma data, é identificada em azul. De fato, o armazenamento de datas na planilha é feito através de uma sequência numérica. Uma formatação de data é aplicada apenas para a apresentação do valor.

Eliminação do apóstrofo antes de números em células

O Realce de valor permite ao usuário identificar os tipos de conteúdo da célula. Essa identificação é fundamental para evitarmos erros de contabilização em fórmulas. A razão é que, eventualmente, conteúdos de células que parecem números são, na verdade, textos.

O Calc é rígido na interpretação desses conteúdos. Por exemplo, numa fórmula de SOMA, como abaixo:

C	
1	
2	
3	
4	
5	
=SOMA(C2:C6)	
, , , , , , , , , , , , , , , , , , , ,	

O que parece um procedimento muito comum pode gerar um resultado confuso se os conteúdos e formatações de célula não forem aplicados da forma correta. O resultado da fórmula de soma, que deveria ser 15, é 12.


С	
1	
2	
3	
4	
5	
12	

Ao aplicarmos o **Realce de valor**, podemos observar que nem todos os conteúdos da lista de números estão sendo interpretados como números. O número 3 está em preto, como se fosse um texto.

C	
1 1	
_	
2	
_	
3	
S	
4	
5	
)	
12	
12	
1	

A razão pode ser variada: uma cópia de conteúdo da Web ou de alguma outra aplicação ou, também, a aplicação equivocada de uma formatação sobre a célula.

Ao clicarmos para editar o conteúdo desta célula, observamos que o número 3 é precedido por um apóstrofo. Na verdade, não é um erro. O apóstrofo pode ser utilizado sempre que o usuário desejar que um conteúdo numérico seja apresentado como um número mas não seja contabilizado em fórmulas. É um recurso existente em praticamente todos os aplicativos de planilhas eletrônicas do mercado.




A eliminação do apóstrofo corrige a interpretação do número 3 e faz com que a fórmula de soma resulte, então, em 15.

С	
1	
3	
3	
4	
5	
15	

Em geral, ocorrências isoladas do apóstrofo em conteúdos numéricos podem ser resolvidas com a edição simples do conteúdo, como fizemos acima. No entanto, quando a correção envolve centenas de células, o procedimento manual é impraticável.


A solução é utilizarmos a função **Localizar e substituir** do menu **Editar**. No campo **Localizar** inserimos **^.** e no campo **Substituir** inserimos **&**. Devemos, também, marcar a opção **Expressões regulares**. Depois, basta clicar em **Substituir todos** para finalizar a correção.



Séries de preenchimento


Uma **Série de preenchimento** é uma forma fácil de fazer um preenchimento automático em uma área da planilha a partir de um valor inicial.

Inicialmente, digite o valor inicial em uma célula. Com a célula selecionada, coloque o ponteiro do mouse sobre o ponto preto no canto inferior direito, chamado **Alça de preenchimento**, até que este se transforme em uma pequena cruz.



Arraste com o botão do mouse pressionado até a última célula da sequência desejada, como no passo 1 apresentado na tabela abaixo. Solte o botão do mouse e a área selecionada será preenchida com a sequência numérica correspondente (passo 2).

Se a direção da sua seleção for horizontal para a esquerda ou vertical para cima, o Calc fará o preenchimento com decremento 1. Se a direção da sua seleção for horizontal para a direita ou vertical para baixo, o Calc fará o preenchimento com incremento 1.



Se você desejar criar uma sequência de preenchimento sem incremento algum, faça o mesmo procedimento, no entanto, ao clicar e arrastar com a alça de preenchimento, mantenha a tecla **Ctrl** pressionada.

Fixação de referências de célula

Ao utilizarmos séries de preenchimento a partir de fórmulas com referências de célula, podemos fixar a referência a um endereço através da combinação de teclas **Shift+F4**.

Essa combinação alterna entre a fixação de linhas e colunas através da inclusão de um símbolo \$ antes da linha ou da coluna a ser fixada. O endereço de célula que possuir o \$ não será incrementado quando o usuário selecionar o intervalo a ser preenchido através da alça de preenchimento.

Detetive

Para descobrirmos visualmente os operandos que compõe a fórmula em uma célula, utilizamos as funções do Detetive, disponíveis no menu **Ferramentas > Detetive**.


Em Rastrear precedentes, verificamos os operandos de uma fórmula selecionada.

Em **Rastrear dependentes**, verificamos em qual fórmula o conteúdo selecionado funciona como um operando.


Para removermos os rastros de uma célula, basta posicionarmos sobre ela e clicarmos no item **Remover precedentes** ou no item **Remover dependentes**. Para removermos os rastros de todas as fórmulas, basta clicarmos em **Remover todos os rastros**.

Os rastros de precedentes e dependentes são apresentados na cor azul se os operandos estiverem corretos. No exemplo abaixo, temos, na célula C6, a fórmula =B4/D4 e, na célula E8, a fórmula =C6+F6.

Sobre ambas foi aplicado o rastreamento de precedentes. Note, no entanto, que o rastreamento de precedentes da célula E8 em relação à célula C6 está indicado em vermelho. A razão é que o resultado da fórmula em C6 está gerando o erro apresentado na célula E8, por isso, esse operando está destacado para identificar a origem do problema.




Se aplicarmos, sobre a célula E8, apenas o rastreamento de erros (menu **Ferramentas > Detetive > Rastrear erro**) identificaremos todas as células que possuem relação com o erro na fórmula da célula.



Atingir meta

O recurso Atingir meta do LibreOffice Calc serve para descobrirmos um valor de uma variável em uma fórmula, a partir de um resultado fornecido. Pode ter muita utilidade principalmente em cálculos matemáticos e financeiros. Tomemos o seguinte exemplo:



Onde as células possuem o seguinte conteúdo:

A2 - Número


A3 - = "Raiz Quadrada de " & B2 & ":"

B2 – 16 (um número qualquer)

B3 - = RAIZ(B2)

Ou seja, temos uma fórmula que calcula a raiz quadrada de um determinado número. Digamos, no entanto, que a nossa necessidade seja descobrir um número a partir da sua raiz quadrada. Sem reescrever a fórmula ou alterar qualquer célula da planilha, podemos descobrir o resultado que gueremos.

Para isso, usaremos a função **Ferramentas > Atingir meta**. Clicando sobre o menu, será aberto o seguinte diálogo:




Onde temos os campos:

Célula de fórmula, que corresponde ao local onde está a fórmula cujo resultado final já sabemos e que contém uma célula variável que queremos descobrir o valor.

Valor desejado, é o resultado final da fórmula, que já devemos conhecer.


Célula variável, é a célula que contém a variável que queremos descobrir. No nosso exemplo, é um número do qual já sabemos a raiz quadrada.

Para o nosso exemplo, teremos, então:



Onde o campo **Célula de fórmula** contém a célula da fórmula da raiz quadrada (B3), o campo **Valor desejado** contém o valor 4,5, que é o valor da raiz conhecida e o campo **Célula variável** contém o valor da célula B2, que conterá a variável que dá origem ao resultado dessa fórmula.

Clicando em **OK**, o LibreOffice informará o resultado da operação e perguntará se o valor calculado deverá ser inserido na célula:



Clicando em Não, os valores não são atualizados.

Clicando em Sim, os valores são atualizados na planilha, como abaixo:

	А	В
1		
2	Número:	20,25
3	Raiz Quadrada de 20,25:	4,5

Temos, por fim, o valor que dá origem ao resultado desejado.

Funções de Banco de Dados

BDSOMA

A função BDSOMA tem como objetivo somar valores correspondentes dentro de um intervalo aos critérios fornecidos pelo usuário. A sintaxe da função é:

=BDSOMA(INTERVALO_DE_PESQUISA; NOME_DA_COLUNA_DA_SOMA; CRITÉRIOS)

Onde:

INTERVALO_DE_PESQUISA é o intervalo onde será feita a avaliação dos critérios e onde está, também, a coluna dos valores a serem somados.

NOME_DA_COLUNA_DA_SOMA é o nome da coluna, dentro do intervalo, que deverá ser somada a partir dos critérios.

CRITÉRIOS é um intervalo de células com a mesma estrutura do INTERVALO_DE_PESQUISA, contendo os argumentos para identificar os valores a serem somados.

Por exemplo, considere a planilha de Despesas abaixo:

	Α	В	С
1	Despesa	Data	Valor
2	Aluguel	10/05	R\$ 500,00
3	Combustível	05/05	R\$ 30,00
4	Cinema	19/05	R\$ 45,00
5	Combustível	16/05	R\$ 50,00
6	Combustível	03/05	R\$ 150,00
7	Condomínio	10/05	R\$ 120,00

Podemos utilizar a função BDSOMA para responder questões como: qual a soma dos gastos realizados na despesa Aluguel e no dia 16/05?

O primeiro passo é construirmos a estrutura dos critérios, que será similar ao intervalo de avaliação original:

	F	G	Н
5	Critérios:		
6	Despesa	Data	Valor
7	Aluguel		
8		16/05	

Note que somente a despesa Aluguel e a data 16/05 foram inseridas nos critérios. A coluna Valor do critério não possuirá preenchimento.

Depois, basta criar a função BDSOMA, indicando a coluna Valor como a coluna a ser somada:

```
=BDSOMA(A1:C7;"Valor";F6:H8)
```

O resultado da função será 550,00 que é a soma do valor da despesa Aluguel (500,00) com o valor gasto no dia 16/05 (50,00).

Como você pode notar, a função BDSOMA assemelha-se muito à função SOMASE. A diferença é que a função BDSOMA permite a inclusão de mais do que um único argumento nos critérios da fórmula.

BDCONTAR

Outra função de banco de dados muito útil é a BDCONTAR. A função é similar a função BDSOMA, com a diferença de que, agora, é feita a contagem da quantidade de registros que obedecem ao critério desejado.

O formato da função é:

=BDCONTAR(INTERVALO_DE_PESQUISA; NOME_DA_COLUNA_DA_CONTAGEM; CRITÉRIOS)

Se considerarmos a planilha de Despesas já apresentada, podemos utilizar a função BDCONTAR para responder questões como: quantas vezes no período foi gasto mais do que 50,00 reais em Combustível?

Nesse caso, os critérios seriam:

	F	G	Н
5	Critérios:		
6	Despesa	Data	Valor

A fórmula BDCONTAR seria, então:

=BDCONTAR(A4:C19; "Valor"; F6:H7)

Cujo resultado é 2, correspondente aos gastos de 50,00 e 150,00 reais em combustível.

Funções de Data e Hora

ANO

Retorna o ano de uma data fornecida. O formato da função é:

```
=ANO(DATA)
```

Onde Data é uma data qualquer entre "aspas duplas" ou um endereço de uma célula que contenha uma data.

```
=ANO("19/12/1970")
```

O resultado da fórmula acima com a função ANO será 1970.

AGORA

A função AGORA() retorna a data e a hora atual do sistema.

Se, por exemplo, hoje é o dia 08/09/2010 e, no momento, são 09:25:10, a função

```
=AGORA()
```

retornará 08/09/10 09:25.

Toda vez que o arquivo é aberto ou que o usuário clica em F9 (função Recalcular) a função AGORA é recalculada. O resultado da função pode ser formatado através do menu Formatar > Células.

DIA

Retorna o dia de uma data fornecida. O formato da função é:

```
=DIA(DATA)
```

Onde Data é uma data qualquer entre "aspas duplas" ou um endereço de uma célula que contenha uma data.

```
=DIA("19/12/1970")
```

O resultado da fórmula acima com a função DIA será 19.

DIATRABALHOTOTAL

Em versões do antigo OpenOffice.org, o nome da função DIATRABALHOTOTAL era DIASÚTEISTOTAIS. A partir da versão 3.1 passou a ser utilizada a nova nomenclatura, também utilizada no LibreOffice. O formato da função, no entanto, continuou o mesmo.

```
=DIATRABALHOTOTAL(DATA_INICIAL; DATA_FINAL; FERIADOS)
```

Onde:

DATA INICIAL é a data a partir do qual os dias úteis serão contados;

DATA FINAL é a data até onde os dias úteis serão contados.

FERIADOS é um intervalo de células onde serão indicadas as datas que não devem ser contabilizadas na contagem.

Ou seja, a função DIATRABALHOTOTAL conta os dias úteis entre a data inicial e final, descontados os sábados, os domingos e os feriados indicados pelo usuário.

Um exemplo interessante da função permite encontrarmos os dias de trabalho em cada mês do ano. Note que, inicialmente, definimos três intervalos. O intervalo de feriados, que é preenchido conforme as datas que identificaremos previamente, o intervalo de datas de início, que corresponde ao primeiro dia de cada mês e o intervalo dos últimos dias de cada mês, calculado a partir da fórmula

=FIMMÊS(DATA_INICIAL;0)

A coluna Dias úteis é, por fim, obtida pelo cálculo da função DIATRABALHOTOTAL com os argumentos definidos para cada mês do ano. Ao final, podemos somar os resultados que teremos o número total de dias trabalhados no ano.

2				
3	Feriados	Início	Fim	Dias úteis
4	01/01/09	01/01/09	31/01/09	=DIATRABALHOTOTAL(C4;D4;A\$4:A\$11)
5	24/02/09	01/02/09	28/02/09	19
6	07/09/09	01/03/09	31/03/09	22
7	21/04/09	01/04/09	30/04/09	21
8	01/05/09	01/05/09	31/05/09	20
9	15/11/09	01/06/09	30/06/09	22
10	02/11/09	01/07/09	31/07/09	23
11	25/12/09	01/08/09	31/08/09	21
12		01/09/09	30/09/09	21
13		01/10/09	31/10/09	22
14		01/11/09	30/11/09	20
15		01/12/09	31/12/09	22
16				254

DOMINGODEPÁSCOA

Retorna a data do domingo de páscoa a partir de um ANO inserido como argumento. O formato da função é:

=DOMINGODEPÁSCOA(ANO)

=DOMINGODEPÁSCOA(1989) resulta em 26/03/89.

ÉANOBISSEXTO

A função ÉANOBISSEXTO apresenta como resultado o valor VERDADEIRO (1), se o ano da data inserida como argumento for um ano bissexto, ou FALSO (0), se o ano da data inserida como argumento não for um ano bissexto.

=ÉANOBISSEXTO (DATA)

Por exemplo:

- =ÉANOBISSEXTO(C5) retorna valor 0 quando a célula C5 possuir a data 01/01/1990.
- =ÉANOBISSEXTO(C5) retorna valor 1 quando a célula C5 possuir a data 01/01/1996.

FIMMÊS

A função FIMMÊS possui a seguinte sintaxe:

```
=FIMMÊS(DATA_REFERÊNCIA; MESES)
```

Retorna a data do último dia do mês indicado pelo número de MESES a partir da DATA REFERÊNCIA.

MESES pode ser um número negativo, se quisermos a data final N meses antes da DATA_REFERÊNCIA, ou um número positivo, se quisermos a data final N meses depois da DATA REFERÊNCIA.

Por exemplo, se quisermos saber a data final do mês de fevereiro de 2008, podemos utilizar:

```
=FIMMÊS("01/02/2008";0)
```

Cujo resultado é: 29/02/08

Se quisermos saber a data do final do mês seis meses depois da data atual, usamos:

```
=FIMMÊS(HOJE();6)
```

A função HOJE() retorna a data do dia atual e 6 representa o número de meses após a data de hoje.

Outro exemplo possível é descobrirmos a data de pagamento conforme a definição a seguir: "... o pagamento será efetuado no último dia do mês subsequente à assinatura do contrato". Supondo que a célula C5 contenha a data de assinatura do contrato, teríamos a data de pagamento definida pela seguinte fórmula FIMMÊS:

```
=FIMMÊS(C5;1)
```

Considerando a data em C5 igual a 22/06/10, o resultado da função será 31/07/10.

HOJE

A função HOJE() retorna a data atual do sistema.

Se, por exemplo, hoje é o dia 08/09/2010, a função =HOJE() retornará 08/09/10.

Toda vez que o arquivo é aberto ou que o usuário clica em F9 (função Recalcular) a função HOJE é recalculada. O resultado da função pode ser formatado através do menu Formatar > Células.

MÊS

Retorna o mês de uma data fornecida. O formato da função é:

```
=MÊS(DATA)
```

Onde Data é uma data qualquer entre "aspas duplas" ou um endereço de uma célula que contenha uma data.

```
=MÊS("19/12/1970")
```

O resultado da fórmula acima com a função MÊS será 12.

Funções Estatísticas

DESVPAD

A função DESVPAD é bastante utilizada em cálculos estatísticos e calcula o desvio padrão de uma amostra. Possui o formato:

=DESVPAD(ARGUMENTOS)

Onde ARGUMENTOS é uma lista de valores numéricos, células ou intervalos de células que representa a amostra a ser calculada.


No exemplo abaixo, calculamos o valor da função DESVPAD sobre as notas obtidas pelos alunos. O resultado final da função é 1,75.

2					
3	Disciplina	Português			
4	Semestre	3			
5	Professora	Caroline Mad	nado		
6	Nota mínima para aprovação:	6			
7					
8					
9	Aluno	Nota			
10	Ana Cláudia <u>Schmidt</u>	8,3			
11	Ana Pereira Gomes	7,1		Média da Turma:	
12	Artur Almada	9,1		7,19	
13	Éverton <u>Brenner</u> Oliveira	5,4			
14	Helena Martins da Silva	9,5		Desvio Padrão:	
15	Jorge Luís Porto	6,9		=DESVPAD(B10:B1	8)
16	Lauro <u>Minetto Bianchini</u>	8			
17	Rodrigo Vargas Neves	4,1			
18	Sandra <u>Susin</u>	6,3			
19					

MAIOR

A função MAIOR retorna o maior valor na enésima posição de um intervalo de células definido. =MAIOR(INTERVALO; POSIÇÃO)

INTERVALO é um intervalo de células válido e POSIÇÃO é a posição do valor desejado em uma ordenação decrescente. No exemplo abaixo, obtemos os valores das três maiores notas usando a função MAIOR:



É interessante salientar a diferença entre a função MAIOR e a função MÁXIMO. A função MAIOR permite a flexibilidade de definirmos a posição na ordem de classificação do intervalo enquanto a função MÁXIMO retorna apenas o maior valor no intervalo.


Funções Estatísticas 21

MÁXIMO

Retorna o valor máximo encontrado dentro de um ou mais intervalos de células definidos como argumentos da função. Possui o formato:

=MÁXIMO(ARGUMENTOS)


No exemplo abaixo, calculamos a nota máxima do intervalo de notas da primeira avaliação. O resultado será, para o intervalo de B11:B19, a nota 9,5.



MÉDIA

A função MÉDIA calcula a média de um intervalo de valores. A estrutura da função MÉDIA é: =MÉDIA (INTERVALO_DE_VALORES)

O intervalo de valores pode ser composto por intervalo de células ou números. No exemplo abaixo, veja que a média das notas dos alunos é obtida pela fórmula =MÉDIA(B11:B19), cujo resultado será 7,32.




MENOR

A função MENOR retorna o menor valor na enésima posição de um intervalo de células definido.

=MENOR(INTERVALO; POSIÇÃO)

INTERVALO é um intervalo de células válido e POSIÇÃO é a posição do valor desejado em uma ordenação crescente. No exemplo abaixo, descobrirmos os três melhores tempos de resposta (medidos em segundos) de uma lista de testes:



Nos casos onde há ocorrências de zero no intervalo de células que devem ser evitados na contabilização, usamos a função CONT.SE com a função MENOR. Abaixo, nosso intervalo de células é o intervalo L3:L20.

=MENOR(L3:L20;CONT.SE(L3:L20;0)+1)


Com a função CONT.SE, obtemos o número total de zeros existentes no intervalo. Somando uma unidade, temos a posição do menor valor do intervalo.

MÍNIMO

Retorna o valor mínimo encontrado dentro de um ou mais intervalos de células definidos como argumentos da função. Possui o formato:

=MÍNIMO(ARGUMENTOS)

No exemplo abaixo, calculamos a nota mínima do intervalo de notas da primeira avaliação. O resultado será, para o intervalo de B11:B19, a nota 4,1.



Funções Estatísticas 23

Funções de Informações

ÉERROS

A função ÉERROS retorna VERDADEIRO caso o argumento avaliado seja um erro ou retorna FALSO caso o argumento avaliado seja um resultado válido. Seu formato é:

=ÉERROS (ARGUMENTO)

Por exemplo, podemos avaliar o resultado de uma divisão. Imaginando um cálculo como 1/0, sabemos, antecipadamente, que o resultado será o erro #DIV/0! (divisão por 0). Podemos utilizar esse cálculo como argumento na função ÉERROS e verificar o resultado VERDADEIRO para a operação:

=ÉERROS(1/0) resulta em VERDADEIRO.

Da mesma forma =ÉERROS(1/1) resulta em FALSO, pois 1/1 é uma operação válida.

Vale destacar que o argumento da função também poderá ser uma referência de célula onde a operação ou valor a ser avaliado está inserido.

É.NÃO.DISP

A função É.NÃO.DISP() possui a seguinte estrutura:

=É.NÃO.DISP(VALOR)

Onde VALOR é um resultado de uma fórmula ou um endereço de célula que contém o valor a ser avaliado. Se VALOR contém o código de erro "#N/DISP", então a função É.NÃO.DISP retorna VERDADEIRO. Se VALOR contém um resultado diferente do código de erro "#N/DISP", então a função É.NÃO.DISP retorna FALSO.

A função É.NÃO.DISP é muito utilizada para a avaliação dos resultados de fórmulas com as funções PROCV. No caso, um resultado "#N/DISP" da função PROCV identifica que o argumento procurado não foi encontrado. Logo, podemos desenvolver uma avaliação da seguinte maneira, considerando o exemplo abaixo, onde fornecemos uma matrícula na célula amarela e obtemos a respectiva nota na célula azul (através da função PROCV):

	A	В	С	D	Е	F	G	Н	I
2									
3	Disciplina	Português							
4	Semestre	3							
5	Professora	Caroline Machado							
6	Média	6							
7									
8							Matrícula	123-5	
9	Matrícula	Aluno	Nota	Resultado					
10	123-5	Ana Cláudia Schmidt	8,3	APROVADO			Nota	8,3	
11	124-2	Ana Pereira Gomes	7,1	APROVADO					
12	125-4	Artur Almada	9,1	APROVADO					
13	126-4	Éverton Brenner Oliveira	5,4	Exame					
14	127-2	Helena Martins da Silva	9,5	APROVADO					
15	128-8	Jorge Luís Porto	6,9	APROVADO					
16	129-1	Lauro Minetto Bianchini	8	APROVADO					
17	130-8	Rodrigo Vargas Neves	4,1	Exame					
18	131-9	Sandra Susin	6,3	APROVADO					
19									

=SE(É.NÃO.DISP(PROCV(H8;A10:D18;3;0)); "Valor não encontrado"; PROCV(H8;A10:D18;3;0))

No exemplo, se o conteúdo de H8 (célula da Matrícula, em amarelo) não é encontrado pela função PROCV na primeira coluna do intervalo de A10:D18, então é retornado o valor "#N/DISP". Por consequência, a função É.NÃO.DISP retorna VERDADEIRO e, então, a função SE define o resultado final como "Valor não encontrado".

Caso o conteúdo de H8 seja encontrado pela função PROCV na primeira coluna do intervalo de A10:D18, então é retornado o valor indicado pela coluna de índice 3 (terceiro argumento do PROCV). Por consequência, a função É.NÃO.DISP retorna FALSO e, então, a função SE define o resultado final como o próprio resultado da função PROCV.

Funções Lógicas

SE

Esta é uma função bastante interessante pois permite ao usuário da planilha construir expressões condicionais, avaliando e apresentando diferentes resultados conforme uma cláusula avaliada.

A estrutura da função SE é:

```
=SE (CONDIÇÃO; VALOR_SE_CONDIÇÃO_VERDADEIRA; VALOR_SE_CONDIÇÃO_FALSA)
```

O primeiro argumento é a condição. Normalmente, avaliamos o conteúdo de uma célula em relação a um dado parâmetro, como, por exemplo C4<100 ou A1="APROVADO".

Caso a condição seja verdadeira, o segundo argumento é apresentado como resultado da função.

Caso a condição seja falsa, o terceiro argumento é apresentado como resultado.

No exemplo abaixo, avaliamos o valor da nota do aluno e, caso esteja acima da média, apresentamos o resultado "Aprovado". Senão, apresentamos o resultado "Recuperação".

2					
3	Disciplina	Português			
4	Semestre	3			
5	Professora	Caroline Machado			
6	Média	6			
7					
8					
9	Aluno	Nota	Resultado		
10	Ana Cláudia Schmidt	8,3	=SE (B10>=B\$6;"Ap	provado";"Recu	uperação")
11	Ana Pereira Gomes	1	Recuperação		
12	Artur Almada	9,1	Aprovado		
13	Éverton Brenner Oliveira	5,4	Recuperação		
14	Helena Martins da Silva	9,5	Aprovado		
15	Jorge Luís Porto	6,9	Aprovado		
16	Lauro Minetto Bianchini	8	Aprovado		
17	Rodrigo Vargas Neves	4,1	Recuperação		
18	Sandra Susin	6,3	Aprovado		
10					

Funções Matemáticas

ABS

Retorna como resultado o valor absoluto do número fornecido.

=ABS(NÚMERO)

Exemplos:

=ABS(120) resulta em 120.

=ABS(-92,22) resulta em 92,22.

ALEATÓRIO

A função ALEATÓRIO retorna um número qualquer entre 0 e 1 como resultado. A função não possui parâmetros e é utilizada na forma:

```
=ALEATÓRIO()
```

É importante salientar que qualquer modificação indireta na célula pode resultar no novo cálculo da função aleatório como, por exemplo, uma mudança de formatação ou a função Ferramentas > Recalcular (F9).

ARRED

Arredonda um número para o valor mais próximo até uma quantidade de dígitos definida pelo usuário.

```
=ARRED(NÚMERO; QUANTIDADE_DE_DIGITOS)
```

Essa função apresenta como resultado o NÚMERO fornecido como primeiro argumento arredondado com a QUANTIDADE_DE_DÍGITOS colocada no segundo argumento, como em:

=ARRED(2,348;2) cujo resultado é 2,35.

Em alguns casos, é necessário mudar o formato da célula para ver todas as decimais. Por exemplo:

=ARRED(-32,4834;3) retorna -32,483 (com a formatação mostrando mais casas decimais).

Se a QUANTIDADE_DE_DÍGITOS for omitida ou for zero, a função arredonda para o inteiro mais próximo:

=ARRED(2,348;0) retorna 2.

Se a QUANTIDADE_DE_DÍGITOS for negativa, a função arredonda para a dezena, centena ou milhar, etc... mais próximo.

=ARRED(835,65;-2) retorna 800.

ARREDONDAR.PARA.BAIXO

Arredonda um número para baixo até uma quantidade de dígitos nas casas decimais definida pelo usuário.

=ARREDONDAR.PARA.BAIXO(NÚMERO; QUANTIDADE_DE_DIGITOS)

Por exemplo:

=ARREDONDAR.PARA.BAIXO(1,234;2) retorna 1,23.

Funções Matemáticas 27

Se a QUANTIDADE_DE_DÍGITOS for omitida ou for zero, a função arredonda para o inteiro mais baixo: =ARREDONDAR.PARA.BAIXO(45,67;0) retorna 45.

Se a QUANTIDADE_DE_DÍGITOS for negativa, a função arredonda para a dezena, centena ou milhar, etc... mais baixa.

=ARREDONDAR.PARA.BAIXO(975,65;-2) retorna 900.

ARREDONDAR.PARA.CIMA

Arredonda um número para cima até uma quantidade de dígitos nas casas decimais definida pelo usuário.

=ARREDONDAR.PARA.CIMA(NÚMERO; QUANTIDADE_DE_DIGITOS)

Por exemplo:

=ARREDONDAR.PARA.CIMA(1,2345;1) retorna 1,3.

Se a QUANTIDADE_DE_DÍGITOS for omitida ou for zero, a função arredonda para o inteiro mais alto: =ARREDONDAR.PARA.CIMA(45,67;0) retorna 46.

Se a QUANTIDADE_DE_DÍGITOS for negativa, a função arredonda para a dezena, centena ou milhar, etc... mais alta.

=ARREDONDAR.PARA.CIMA(975,65;-2) retorna 1000.

CONT.NÚM


A função CONT.NÚM conta quantos valores numéricos estão entre os ARGUMENTOS da função. Entende-se como valores numéricos: números, datas e fórmulas cujo resultado seja um número. Células vazias ou células com conteúdo de texto não são contadas na função CONT.NÚM.

O formato da função é:

=CONT.NÚM(ARGUMENTOS)

Observe no exemplo abaixo que nem todos os alunos fizeram a primeira avaliação. Podemos usar a função CONT.NÚM para contar as notas do intervalo B11:B19 e identificar quantos alunos de fato fizeram a prova.

O resultado da função =CONT.NÚM(B11:B19) será 7 pois, as duas células, correspondentes aos alunos que não fizeram a prova, estão vazias.



Funções Matemáticas 29

CONT.SE

A função CONT.SE tem como objetivo contar quantos valores obedecem a um determinado critério. A estrutura é bastante simples:

=CONT.SE (INTERVALO; CONDIÇÃO)

Os valores dentro do intervalo são avaliados um a um de acordo com a condição. O valor é contado somente se a condição for verdadeira.

No exemplo abaixo, contamos quantos alunos estão com notas acima da média estabelecida. Note que usamos uma concatenação de texto para expressar adequadamente o critério, indicado no segundo argumento com a expressão ">"&B6. Ou seja, concatenamos o sinal de > com o conteúdo da célula B5.

2							
3	Disciplina	Português					
4	Semestre	3					
5	Professora	Caroline Mac	hado				
6	Média	6					
7							
8							
9	Produto	Valor		Quantidade	de alunos com	notas acima	da média:
10	Ana Cláudia Schmidt	8,3		=CONT.SE(B1	.0:B18;">"&B6)		
11	Ana Pereira Gomes	7,1					
12	Artur Almada	9,1					
13	Éverton Brenner Oliveira	5,4					
14	Helena Martins da Silva	9,5					
15	Jorge Luís Porto	6,9					
16	Lauro Minetto Bianchini	8					
17	Rodrigo Vargas Neves	4,1					
18	Sandra Susin	6,3					
10							

O resultado da função CONT.SE acima é de 7 alunos.

Quando o critério ou condição for de igualdade, não precisamos usar a concatenação de texto, por exemplo:

=CONT.SE(B10:B18;B6)

Em sua construção mais comum, a função CONT.SE permite apenas um argumento como critério de contagem. No entanto, em casos específicos, é possível utilizar mais de um argumento através do uso de expressões regulares em fórmulas.

O exemplo abaixo ilustra essa situação. A partir da tabela abaixo, desejamos obter a quantidade de cadastros de pessoas que são dos estados do Rio Grande do Sul e Santa Catarina.



Uma abordagem óbvia e simples seria a soma de CONT.SE:

```
=CONT.SE(F3:F8; "RS")+CONT.SE(F3:F8; "SC")
```

Uma abordagem elegante poderia utilizar expressões regulares:

```
=CONT.SE(F3:F8; "RS|SC")
```

Onde o símbolo | (pipe) entre as siglas RS e SC representa a operação OU lógica. Logo, estamos contando apenas os valores do intervalo de F3 até F8 que são iguais a RS ou a SC.

CONT.VALORES


A função CONT.VALORES permite contar células preenchidas com valores de texto, número ou fórmula dentro de um intervalo.

O formato da função CONT.VALORES é:

=CONT.VALORES(ARGUMENTOS)

No exemplo abaixo, o usuário deverá preencher o espaço amarelo com cinco códigos de produto. Para contar quantas células já foram preenchidas, utilizamos a função CONT.VALORES e o intervalo de G5:G9.

O resultado da função, no exemplo abaixo, será 3.




Funções Matemáticas 31

CONTAR.VAZIO

CONTAR.VAZIO conta quantas células do intervalo indicado em ARGUMENTOS estão vazias, isto é, sem conteúdo algum.

=CONTAR.VAZIO(ARGUMENTOS)

No nosso exemplo anterior, da função CONT.VALORES, calculamos quantas células do intervalo amarelo já haviam sido preenchidas. Podemos encontrar a informação complementar (quantas células faltam ser preenchidas) através da função CONTAR.VAZIO.



No exemplo, a função CONTAR.VAZIO pode ser utilizada sobre o intervalo de G5:G9. O resultado da função na célula H15 será de 2.

SINAL

A função SINAL identifica se um número fornecido como argumento é positivo ou negativo. =SINAL (ARGUMENTO)

Se o número for positivo, o resultado da função será o número 1. Se for negativo, o resultado da função será -1. Caso o número testado seja 0, o resultado da função será 0.

Um exemplo do uso da função SINAL é na operação de valores contábeis. Na figura abaixo, a coluna Operação contém a fórmula SINAL para todos os valores da coluna Transações. Conforme o tipo de transação (entrada ou saída), o resultado da operação é 1 (valores positivos) ou -1 (valores negativos).

D8			▼ fω Σ = □	=SINAL(C8)
	Α	В	С	D
2				
2			Entradas:	712,28
3			Saídas:	264,77
4				
5			Transações (R\$)	Operação
6			15,43	1
7			54,54	1
8			(65,22)	-1
9			15,22	1
10			54,33	1
11			(67,55)	-1
12			102,33	1
13			160,00	1
14			(132,00)	-1
15			210,00	1
16			25,00	1
17			75,43	1
10			·	

A partir dos resultados da coluna Operação, é possível contabilizar os totais de entradas e saídas através de fórmulas SOMASE, respectivamente nas células D2 [fórmula =SOMASE(D6:D17;1;C6:C17)] e D3 [fórmula =ABS(SOMASE(D6:D17;-1;C6:C17))].

Na fórmula da célula D3 utilizamos, também, a função ABS, que retorna o valor absoluto de um dado número.

Funções Matemáticas 33

SOMASE

A função SOMASE é útil para cálculos que envolvam valores totais a partir de um determinado critério. O formato da função é:

=SOMASE(INTERVALO_DE_AVALIAÇÃO; CRITÉRIO; INTERVALO_DE_SOMA)

Os valores do intervalo de avaliação são avaliados conforme o critério. Caso estejam de acordo com o critério indicado, o valor correspondente no intervalo de soma é somado ao resultado.



O resultado da fórmula SOMASE acima é de R\$ 70,00.

A função SOMASE possui uma diferença significativa em relação à sua correspondente no Microsoft Excel. No Calc, o formato do CRITÉRIO deve ser equivalente ao formato dos conteúdos das células no INTERVALO_DE_AVALIAÇÃO. Ou seja, se os conteúdos do INTERVALO DE AVALIAÇÃO são textos, então o CRITÉRIO também deverá ser um texto.

Na figura abaixo essa situação pode ser observada. Os valores correspondentes à categoria (coluna E), apesar de serem números, foram formatados como texto e a célula que contabiliza a soma de valores (I3), contém a função =SOMASE(E2:E5;323;F2:F5), onde o valor 323 foi inserido como número.

	D	E	F	G	Н	I	
1		Categoria	Valor				
2		1	1				
3		2	2		Soma dos valores da categoria 323:	0	
4		323	3				
5		323	4				
-							

O resultado para a função, devido à diferença de formatos, é zero. Para resolver a questão devemos ajustar o formato do intervalo de células E2:E5 para número, adequando o formato de célula ao tipo de conteúdo utilizado.

SUBTOTAL

Quando quisermos contabilizar um resultado a partir de um intervalo de células com autofiltro, por exemplo, usamos a função SUBTOTAL com o seguinte formato:

=SUBTOTAL (CÓDIGO_DA_FUNÇÃO; INTERVALO_DE_DADOS)

Onde o CÓDIGO_DA_FUNÇÃO define que função será utilizada para calcular o subtotal. A tabela abaixo define os códigos que podemos utilizar na função:


Código	Função
1	MÉDIA
2	CONT.NÚM
3	CONT.VALORES
4	MÁXIMO
5	MÍNIMO
6	MULT
7	DESVPAD
8	DESVPADP
9	SOMA
10	VAR
11	VARP

A função SUBTOTAL calcula, então, apenas os valores de células visíveis, desconsiderando os valores em células ocultas. Por isso, torna-se uma função interessante para ser utilizada com autofiltros. Considere o exemplo da planilha abaixo:


	A	В	С	D
3				
4	Descrição ±	Conta ±	Data <u>±</u>	Valor <u></u> ±
5	Aluguel	1649-6	10/05	R\$ 500,00
6	Cinema	1342-3	05/05	R\$ 30,00
7	Cinema	1342-3	19/05	R\$ 45,00
8	Combustível	1649-6	16/05	R\$ 50,00
9	Combustível	1649-6	03/05	R\$ 150,00
10	Condomínio	1342-3	10/05	R\$ 120,00
11	Energia Elétrica	1342-3	10/05	R\$ 60,00
12	Padaria	1342-3	29/05	R\$ 15,00
13	Padaria	1649-6	19/05	R\$ 8,00
14	Padaria	1649-6	11/05	R\$ 12,00
15	Restaurante	1342-3	24/05	R\$ 35,00
16	Restaurante	1342-3	10/05	R\$ 62,00
17	Restaurante	1649-6	16/05	R\$ 28,00
18	Supermercado	1649-6	22/05	R\$ 170,00
19	Supermercado	1342-3	03/05	R\$ 15,00
20	Padaria	1649-6	22/05	R\$ 10,00
21	Supermercado	1342-3	16/05	R\$ 25,00
22				

Funções Matemáticas 35

Ao aplicar o autofiltro na coluna A selecionando o critério "Padaria", obteremos apenas a exibição das linhas 12, 13, 14 e 20.




Posicionando o cursor na célula 22 e clicando no botão ∑ da barra de fórmulas, a função SUBTOTAL é inserida na célula 22 com a função SOMA (código 9) e o intervalo D5:D21).



Note que, por padrão, o botão ∑ insere a função SOMA na célula selecionada. A função SUBTOTAL só é utilizada quando o Calc identifica que a área imediatamente acima da célula selecionada possui um autofiltro aplicado.


A função SUBTOTAL também é criada automaticamente quando o usuário faz o cálculo de subtotais através do menu **Dados > Subtotais**. Nesse caso, a operação indicada pelo código da função corresponde à escolha do usuário na lista Utilizar função.




Funções de Planilha

CORRESP

A função CORRESP responde a seguinte pergunta: qual a posição do elemento X num dado vetor de elementos? Por exemplo: qual a posição do elemento Maçã no vetor abaixo?



A resposta é o número 3. Numa planilha do Calc teríamos:



Onde =CORRESP("Maçã";D4:D7;0) tem como resultado o número 3.

A função CORRESP possui a seguinte estrutura:

=CORRESP(CRITÉRIO; INTERVALO; TIPO)

Onde o resultado da função é a posição, dentro do INTERVALO, da célula cujo conteúdo é igual ao CRITÉRIO.

TIPO é um argumento opcional que pode receber os valores -1, 0 e 1. Se o seu valor é igual a 1, a primeira coluna do INTERVALO está em ordem crescente. Se o valor é igual a -1, a primeira coluna do INTERVALO está em ordem decrescente. Se o valor é igual a 0 somente valores exatamente iguais ao critério serão encontrados.

Por exemplo, na tabela abaixo:

	Α
1	Mês
2	Janeiro
3	Fevereiro
4	Março
5	Abril
6	Maio
7	Junho
8	Julho
9	Agosto
10	Setembro
11	Outubro
12	Novembro
13	Dezembro

A fórmula =CORRESP("Abril";A2:A13;0) retorna o valor 4, que é a posição correspondente ao critério "Abril" dentro do intervalo de A2:A13.

Para desabilitarmos a procura por aproximação na função CORRESP, utilizamos o terceiro parâmetro (opcional) com o valor FALSO ou 0. Nesse caso, a função só retornará valores exatos e, caso o valor não exista, o resultado será #N/DISP (Valor não disponível). O resultado #N/DISP pode ser manipulado pela função É.NÃO.DISP().

DESLOC

Retorna o valor do deslocamento de um intervalo por um determinado número de linhas e colunas a partir de um ponto de referência especificado.

```
=DESLOC(CÉL_REFERÊNCIA; LINHAS; COLUNAS; ALTURA; LARGURA)
```

A referência retornada pode ser uma única célula ou um intervalo de células. Você pode especificar o número de linhas e de colunas a serem retornadas de forma a referenciar um intervalo.

Um exemplo interessante é o da planilha abaixo, onde usaremos a função DESLOC combinada com a função CORRESP, vista anteriormente, e a função SOMA.

	Α	В	C
1	Classe	Sexo	Contagem
2	Categoria 1	F	3
3		М	2
4	Categoria 2	F	3
5		М	4
6	Categoria 3	F	5
7		М	4
8	Categoria 4	F	6
9		М	1

Imagine que temos um grupo de pessoas divididas em Classe (Categorias) e Sexo e desejamos saber quantas pessoas existem em uma determinada categoria. Se a categoria desejada é a "Categoria 3" o primeiro passo é descobrir onde ela está posicionada com a função CORRESP:

```
=CORRESP("Categoria 3";A2:A9; 0)
```

Cujo resultado é a posição 5.

Com essa informação, podemos utilizar a função DESLOC a partir da primeira célula preenchida para localizarmos o intervalo de células que contenha os dois valores respectivos aos sexos (F/M) da "Categoria 3".

```
DESLOC(A1;CORRESP("Categoria 3";A2:A9; 0);2;2;1)
```


Onde:

A1 = CÉL REFERÊNCIA do início da definição do intervalo;

CORRESP("Categoria 3";A2:A9; 0) = valor 5, que é o número de linhas a partir do qual será feito o deslocamento.

- 2 = deslocamento de 2 colunas, identificando que a coluna será posicionada sobre a coluna da Contagem.
- 2 = altura de 2 linhas a partir da posição do deslocamento. Essa altura seleciona os números 5 e 4 da "Categoria 3".
- 1 = largura de apenas uma coluna.

Ou seja, você pode imaginar que o primeiro parâmetro é a base para o início do deslocamento, o segundo e o terceiro parâmetros funcionam como o deslocamento propriamente dito e o quarto e o quinto parâmetros servem para a delimitação do tamanho do intervalo.



Nosso resultado até o momento será, então, o intervalo de C6:C7. Finalizaremos o nosso cálculo com a operação final da SOMA aplicada a esse intervalo:

```
=SOMA(DESLOC(A1;CORRESP("Categoria 3";A2:A9; 0);2;2;1))
```

O resultado da soma das contagens da "Categoria 3" será, por fim, 9, que é a soma de 5 pessoas do sexo feminino e 4 pessoas do sexo masculino..

ESTILO

A função ESTILO aplica ou substitui um estilo de célula a célula corrente, durante um determinado período de tempo opcional.

```
=ESTILO (NOME_DO_ESTILO_1; TEMPO; NOME_DO_ESTILO_2)
```

NOME_DO_ESTILO_1 é o estilo aplicado à célula. O nome do estilo deve ser inserido na fórmula entre aspas duplas;

TEMPO é o intervalo em segundos após o qual o ESTILO_1 será substituído pelo ESTILO_2. Esse parâmetro é opcional, ou seja, se não for indicado, não haverá substituição. O tempo é calculado sempre que o arquivo for aberto ou que a função F9 (Recalcular) for acionada;

NOME_DO_ESTILO_2 também é um argumento opcional e representa o estilo que será aplicado á célula em substituição ao ESTILO_1. O nome do estilo deverá ser inserido entre aspas duplas e, caso seja omitido, será considerado o estilo Padrão.

No exemplo abaixo, a função ESTILO substitui o estilo "Amarelo" por "Vermelho" após 120 segundos.

```
=ESTILO ("Amarelo";120, "Vermelho")
```

Note que a função ESTILO é uma função de formatação e não de resultado. Por isso, seu resultado é sempre 0. Para que esse resultado não influencie no seu cálculo, você poderá usar as seguintes estruturas a seguir.

Exemplo da função ESTILO com conteúdos de texto:

```
="Texto da célula."&TEXTO(ESTILO("Padrão");"#")
```

Exemplo da função ESTILO com conteúdos numéricos:

```
=4543,22 + ESTILO("Padrão")
```


Outro exemplo do que pode ser feito com a função ESTILO é muito similar à estrutura de uma formatação condicional quando considerado o valor de alguma outra célula que não a célula onde estará a fórmula. Ou seja, testaremos um valor de uma célula de referência com a função SE e concatenaremos o resultado condicional com a função ESTILO.

Os resultados condicionais serão obtidos dos conteúdos das células da coluna D (D2, D3 e D4), que funcionarão como uma espécie de legenda para a planilha. Da mesma forma, as células da coluna E (E2, E3 e E4) serão utilizadas para que sejam criados os estilos de célula Atenção, Normal e Verificar (consulte a Ajuda do LibreOffice para saber como é possível criar um estilo novo a partir de uma célula).

	A	В	С	D	E
1					
2	Valor	90		Verificar (valores de 0 a 40)	
3				Atenção (valores de 40 a 70)	
4				Normal (valores de 70 a 100)	
5					
6	Resultado	Normal (valores de 70 a 100)		Estilos e formataçã	io 🗵
7					ð 🖢 🖰
8				Atenção Normal	
9				Verificar	
10					
11					
12					
13				Estilos personalizados	

Na célula B6, onde vamos calcular o resultado a partir da avaliação do valor de B2, teremos a seguinte fórmula:

=SE(B2<=40;D2&T(ESTILO("Verificar"));SE(B2<=70;D3&T(ESTILO("Atenção"));D4&T(ESTILO("Normal"))))



Note que a combinação da função ESTILO com a função SE poderá ampliar as possibilidades de uso da formatação condicional. Enquanto a formatação condicional considera apenas três condições, a função SE pode ampliar esse número. Note, no entanto, que essa alternativa só é válida quando a avaliação é feita a partir de uma fórmula que avalia o valor de uma outra célula e não da célula corrente.

ÍNDICE

A função ÍNDICE permite encontrar um valor dentro de um intervalo a partir das referências de linha e coluna desejadas. Sua estrutura mais comum é:

=ÍNDICE(INTERVALO_DE_PESQUISA; LINHA; COLUNA)

Observe o exemplo abaixo:

	A	В	С	D	E	
7						
8			Distâncias entr			
9						
10			Brasília	Florianópolis	São Paulo	
11		Brasília	0	156	178	
12		Florianópolis	156	0	345	
13		São Paulo	178	345	0	
1.4						

Se desejarmos obter uma distância entre duas cidades podemos utilizar a função ÍNDICE. Bastaria utilizarmos como INTERVALO_DE_PESQUISA o intervalo C11:E13 e os índices das cidades desejadas, por exemplo:

=ÍNDICE(C11:E13;1;3)

retornará a distância entre Brasília (correspondente à linha 1) e São Paulo (correspondente à coluna 3). O resultado final da função é 178.

Note ainda que você poderá descobrir os índices relativos às cidades através da função CORRESP. Ou seja, em vez de indicar explicitamente os índices de linha e coluna dentro das fórmulas, podemos usar a função CORRESP para descobri-los de forma mais intuitiva.

Utilizando a célula D3 com o nome da cidade de origem, podemos procurar o conteúdo de D3 no intervalo B11:B13, que nos indicará a linha correta para a função ÍNDICE.

			_		-	_
	A	В	С	D	E	F
2						
3			Origem:	Brasília	=CORRESP(D3;B11:B	313)
4			Destino:	São Paulo	3	
5						
6			Distância:	178		
7						
8			Distâncias entr	e cidades		
9						
10			Brasília	Florianópolis	São Paulo	
11		Brasília	0	156	178	
12		Florianópolis	156	0	345	
13		São Paulo	178	345	0	
14						

Utilizando a célula D4 com o nome da cidade de destino, podemos procurar o conteúdo de D4 no intervalo C10:E10, que nos indicará a coluna correta para a função ÍNDICE.

	A	В	С	D	E
2					
3			Origem:	Brasília	1
4			Destino:	São Paulo	=CORRESP(D4;C10:E10)
5					
6			Distância:	178	
7					
8			Distâncias entr	e cidades	
9					
10			Brasília	Florianópolis	São Paulo
11		Brasília	0	156	178
12		Florianópolis	156	0	345
13		São Paulo	178	345	0
14					

Depois, basta referenciar os valores na função ÍNDICE da célula D6 que nos dará o resultado final:

	A	В	С	D	E
2					
3			Origem:	Brasília	1
4			Destino:	São Paulo	3
5					
6			Distância:	=ÍNDICE(C11:E13;E3;E4)	
7					
8			Distância	s entre cidades	
9					
10			Brasília	Florianópolis	São Paulo
11		Brasília	0	156	178
12		Florianópolis	156 0		345
13		São Paulo	178 345		0
14					

LINHA

A função LINHA não possui argumentos e devolve como resultado o número da linha do endereço da célula corrente.

Um dos usos mais comuns da função LINHA é o de retornar uma ordenação numérica sequencial com base na numeração das linhas. Veja o exemplo abaixo. Note que devido aos campos da parte superior da planilha, a ordenação da coluna Número é feita através da fórmula =LINHA()-7, onde o número sete é a diferença exata para que a numeração inicie em 1 na célula B8.

B8	B8 ▼ 💃 🗵 = =LINHA()-7					
	A	В	С	D		
1			Controle P	rocessual		
2						
3		Setor:	Departamento Fi	nanceiro		
4		Data:	19/03/2010			
5						
6		Manager	DDOTOCOLO.	PPOCEAGO		
7		Número	PROTOCOLO	PROCESSO		
8		1	944798	01375-2005-373-04-00-2		
9		2	929194	00298-2005-024-04-00-9		
10		3	929594	00144-2006-871-04-00-0		
11		4	930492	00113-2007-661-04-00-7		
12		5	941072	01061-2007-733-04-00-5		
13		6	938999	01171-2007-733-04-00-7		
14		7	932147	00541-2006-661-04-00-9		
15		8	941513	00825-2004-611-04-00-7		

Caso uma linha seja adicionada no meio do intervalo de registros, basta copiar a fórmula =LINHA()-7 para que a numeração seja inserida corretamente na nova linha.

PROC


O resultado da função PROC é o conteúdo da célula do intervalo do resultado, correspondente ao conteúdo localizado no intervalo de pesquisa.

O formato da função é:

=PROCV(ARGUMENTO_DE_PESQUISA; INTERVALO_DE_PESQUISA; INTERVALO_DO_RESULTADO)

O resultado da função é o conteúdo da célula do INTERVALO_DO_RESULTADO localizada na mesma posição da célula que contém o ARGUMENTO_DE_PESQUISA no INTERVALO_DE_PESQUISA. É importante salientar que os dois intervalos não precisam ser adjacentes.

Veja o exemplo abaixo:



PROCV

A função PROCV é uma função de procura muito útil. Com ela podemos fazer uma busca de um determinado valor dentro de um intervalo e retornar como resultado um valor de uma coluna adjacente.

A estrutura da função PROCV é a seguinte:

=PROCV (VALOR_PROCURADO; INTERVALO_DE_PESQUISA; ÍNDICE_DA_COLUNA; ORDEM)

O valor procurado é pesquisado dentro da primeira coluna do intervalo de pesquisa. Quando o valor é encontrado, o resultado correspondente, indicado pelo índice da coluna, é apresentado. A ordem é um argumento opcional que pode assumir o valor verdadeiro ou falso. Caso tenha o valor falso, a pesquisa será realizada sempre considerando valores exatos. Por exemplo, podemos fazer uma procura por dados de uma pessoa a partir do seu nome ou do seu número de cadastro. No exemplo abaixo, temos uma tabela com dados de alunos e suas respectivas notas.

8						
9	Matrícula	Aluno	Nota	Resultado	Matrícula	126-4
10	123-5	Ana Cláudia Schmidt	8,3	APROVADO		
11	124-2	Ana Pereira Gomes	7,1	APROVADO	Nome	=PROCV(H9;A10:D18;2;0)
12	125-4	Artur Almada	9,1	APROVADO	Nota	5,4
13	126-4	Éverton Brenner Oliveira	5,4	Exame	Resultado	Exame
14	127-2	Helena Martins da Silva	9,5	APROVADO		
15	128-8	Jorge Luís Porto	6,9	APROVADO		
16	129-1	Lauro Minetto Bianchini	8	APROVADO		
17	130-8	Rodrigo Vargas Neves	4,1	Exame		
18	131-9	Sandra Susin	6,3	APROVADO		
19						

Para fazer uma busca pelos desempenho dos alunos na disciplina, podemos usar a função PROCV. Procurando pelo campo de matrícula na primeira coluna do intervalo, podemos achar os demais dados do aluno. Se desejarmos como resultado o nome do aluno cuja matrícula é 126-4, teríamos a seguinte fórmula: **=PROCV(H9;A10:D18;2;FALSO)**. O resultado seria o nome **Éverton Brenner Oliveira**. Para chegar a esse resultado, a função procurou pela matrícula 126-4 na primeira coluna do intervalo A10:D18. Ao encontrar o registro pesquisado, a função verificou qual o índice da coluna do intervalo A10:D18. O índice, cujo valor é 2, indica a segunda coluna do intervalo. A intersecção entre a linha indicada pelo número de matrícula e o índice da coluna do intervalo indicam o resultado final da fórmula.

Para evitar a pesquisa por aproximação, inserimos o quarto argumento com o valor FALSO. Dessa forma, somente os valores existentes no intervalo retornarão resultados válidos.

Dicas:

- sempre utilizar intervalos ordenados pela primeira coluna;
- usar, na primeira coluna, valores únicos e não nulos.

Funções de Texto

CARACT


A função CARACT apresenta como resultado o caractere correspondente ao número fornecido como argumento da função:

=CARACT(ARGUMENTO)

Letras, números e símbolos da escrita em geral (pontuação, acentos, operações matemáticas) são encontrados a partir do número 33 como ARGUMENTO.

Exemplos:

- =CARACT(97) resulta na letra a.
- ="Resultado: " & CARACT(10) & "12" resulta em:



CONCATENAR

A função CONCATENAR permite agregar, em um único resultado, várias sequências de caracteres referenciados pelos ARGUMENTOS da função.

=CONCATENAR(ARGUMENTOS)

Os ARGUMENTOS podem ser trechos de texto, valores numéricos ou referências de célula. Observe no exemplo abaixo a composição do código do produto usando a função CONCATENAR.

A coluna Código utiliza a função CONCATENAR combinando as referências de célula das colunas Tipo, Série e Número, além do "-", utilizado para a composição do formato correto dos códigos.

Uma observação útil é que o mesmo procedimento pode ser feito com o operador de texto de concatenação, o &. Nesse caso, a fórmula utilizada para uma construção similar à do exemplo acima seria: =B3 & C3 & "-" & D3.

В	С	D	E .
Tipo	Série	Número	Código
CDS	12A	1	=CONCATENAR(B3;C3;"-";D3)
CDS	12A	2	CDS12A-2
CDS	12A	3	CDS12A-3
CDS	12A	4	CDS12A-4
CDS	12A	7	CDS12A-7
CDS	12A	9	CDS12A-9
ASN	13V	2	ASN13V-2
ASN	13V	4	ASN13V-4
ASN	13V	5	ASN13V-5
ASN	13V	6	ASN13V-6
ASN	13V	7	ASN13V-7
ASN	13V	8	ASN13V-8

DIREITA

Devolve como resultado os N últimos caracteres do texto utilizado como ARGUMENTO.

O formato da função é:

=DIREITA (ARGUMENTO; N)

O parâmetro N é opcional. Caso o mesmo seja omitido, o valor retornado pela função corresponderá apenas ao último caractere do texto do ARGUMENTO.

Por exemplo:

=DIREITA("Carro";2) retornará como resultado o texto "ro".

=DIREITA("Carro") retornará como resultado a letra "o".

Funções de Texto 47

ESQUERDA

Devolve como resultado os N primeiros caracteres do texto utilizado como ARGUMENTO a partir da esquerda.

O formato da função é:

```
=ESQUERDA (ARGUMENTO; N)
```

O parâmetro N é opcional. Caso o mesmo seja omitido, o valor retornado pela função corresponderá apenas ao primeiro caractere do texto do ARGUMENTO.

Por exemplo:

=ESQUERDA("LibreOffice";2) retornará como resultado o texto "Li".

=ESQUERDA("LibreOffice") retornará como resultado a letra "L".

EXT.TEXTO

Possui a estrutura:

```
=EXT.TEXTO (ARGUMENTO; POSIÇÃO_INICIAL; TAMANHO)
```

A função EXT.TEXTO retorna um conteúdo de texto obtido a partir de um ARGUMENTO de texto original. O texto resultante é obtido a partir da POSIÇÃO_INICIAL e do TAMANHO, em quantidade de caracteres, definidos no segundo e terceiro argumentos.

Exemplo:

=EXT.TEXTO("LibreOffice";3;2) resulta no texto "br".

TEXTO

Converte um número em um texto conforme o formato especificado.

```
=TEXTO(ARGUMENTO; FORMATO)
```

Onde ARGUMENTO é um número ou uma referência de célula contendo um número e FORMATO é o código de formato do texto que define o formato (equivalente aos códigos de formato da formatação de célula).

VALOR

Converte um texto em um número.

```
=VALOR(ARGUMENTO)
```

Onde ARGUMENTO é um texto ou uma referência de célula contendo um texto.

MAIÚSCULA

Apresenta o ARGUMENTO em letras maiúsculas.

```
=MAIÚSCULA(ARGUMENTO)
```

Por exemplo:

=MAIÚSCULA("Bom dia") retorna BOM DIA.

NÚM.CARACT

A Função NÚM.CARACT apresenta como resultado o tamanho, em número de caracteres, do TEXTO inserido na função como argumento. A estrutura da função é:

=NÚM.CARACT(TEXTO)

O resultado da função =NÚM.CARACT("Teste") é igual a 5.

PRI.MAIÚSCULA

Apresenta o texto do ARGUMENTO com as primeiras letras de cada palavra em maiúsculas.

=PRI.MAIÚSCULA(ARGUMENTO)

Por exemplo:

=PRI.MAIÚSCULA("Bom dia") retorna Bom Dia.

Funções de Texto 49

Funções Matriciais

TRANSPOR

A função TRANSPOR apresenta como resultado a matriz transposta de uma matriz inserida como argumento. Uma matriz transposta faz com que a sequência de valores de linhas passem para colunas e vice-versa.

A função possui a seguinte estrutura:

=TRANSPOR (matriz)

Para usá-la, inicialmente selecione a área onde será colocada a matriz transposta resultante de tamanho equivalente à matriz original. Depois, insira a fórmula transpor tendo como argumento o endereço de células da planilha original.

	A	В	C	D	E
1					
2		1	2	3	
3		4	5	6	
4		7	8	9	
5					3 L x 3 C
6					
7					
8					
9				=TRANSPOR	(B2:D4)
10					

Ao final da digitação da função TRANSPOR com a matriz original como argumento, clique em Ctrl+Shift+Enter para finalizar a função definido-a, explicitamente, como uma função matricial. O resultado será equivalente ao apresentado abaixo:

	A	В		D	
1					
2		1	2	3	
3		4	5	6	
4		7	8	9	
5					
6					
7		1	4	7	
8		2	5	8	
9		3	6	9	
10					

Observe agora que, para qualquer célula clicada no intervalo de B7 até D9, a fórmula será apresentada com chaves: {=TRANSPOR(B2:D4)}. As chaves indicam que a fórmula utiliza matrizes. Caso queiramos editar a fórmula com o assistente de funções, por exemplo, devemos selecionar todo o intervalo. Neste exemplo, de B7 até D9. Se tentarmos editar apenas uma célula do intervalo, receberemos uma mensagem informando que é impossível modificar apenas uma parte do resultado de uma fórmula matricial. Nesse caso, será necessário selecionar todo o intervalo da fórmula para modificação.

SOMARPRODUTO

Multiplica os componentes correspondentes nos vetores indicados como argumentos e retorna a soma destes produtos.

No exemplo abaixo, contabilizamos uma venda a partir do código da mercadoria, do seu valor unitário e da quantidade. Para obtermos o total da venda, teríamos que fazer, para cada mercadoria, a multiplicação (produto) do Valor Unitário pela respectiva Quantidade e, depois, a soma desses resultados.

Podemos utilizar, nesse caso, a função SOMARPRODUTO com a seguinte estrutura:

=SOMARPRODUTO(B2:B4; C2:C4)

Cujo resultado é igual a fazermos a fórmula:

=B2*C2+B3*C3+B4*C4

	A	В	С
1	Código da Mercadoria	Valor Unit.	Quantidade
2	1532-4	R\$ 14,00	3
3	4562-8	R\$ 23,00	2
4	4571-4	R\$ 17,60	1

O resultado da função para o exemplo acima será 105,60.

Outro uso da função SOMARPRODUTO é a contabilização de valores a partir de critérios que devem ser obedecidos por conteúdos de colunas adjacentes. Esse uso é similar ao da função SOMASES, a partir do Excel 2007, e também das fórmulas combinadas SOMA e SE resultantes do uso do assistente de soma das versões antigas do Microsoft Excel.

Para exemplificar esse uso, imagine a planilha abaixo e a seguinte questão: qual a quantidade total de modelos aprovados nos três testes?

	A	В	С	D	E
1					
2		Quantidade	Teste 1	Teste 2	Teste 3
3	Modelo 1	30	Aprovado	Aprovado	Aprovado
4	Modelo 2	50	Aprovado	Aprovado	Reprovado
5	Modelo 3	10	Aprovado	Reprovado	Aprovado
6	Modelo 4	70	Aprovado	Reprovado	Aprovado
7	Modelo 5	50	Aprovado	Reprovado	Aprovado
8	Modelo 6	10	Reprovado	Reprovado	Aprovado
9	Modelo 7	60	Aprovado	Aprovado	Aprovado
10	Modelo 8	80	Aprovado	Reprovado	Reprovado
11	Modelo 9	30	Aprovado	Reprovado	Aprovado
12	Modelo 10	20	Aprovado	Aprovado	Aprovado
13	Modelo 11	40	Aprovado	Reprovado	Reprovado
14	Modelo 12	20	Aprovado	Reprovado	Aprovado
15					

Funções Matriciais 51

Para descobrir a resposta, usamos a fórmula:

```
=SOMARPRODUTO(C3:C14="Aprovado";D3:D14="Aprovado";E3:E14="Aprovado";B3:B14)
```

Onde cada item no intervalo de quantidades (B3:B14) é multiplicado pelo respectivo item nas colunas C3:C14, D3:D14 e E3:E14. O item nessas colunas depende, no entanto, se a operação de comparação com a palavra "Aprovado" é verdadeira ou não. Se for verdadeira, o resultado será 1 para o item, se for falsa, o resultado será 0.

Logo, a quantidade só será contabilizada quando os testes resultarem em Aprovado e o valor da comparação for igual a 1. A multiplicação, nesse caso, será a própria quantidade. Caso o valor seja 0, o resultado da multiplicação pela quantidade será 0, o que não interferirá na soma final.

Como será possível observar, o resultado final desta fórmula será 110. Em outras palavras, o cálculo será similar a:

```
= SOMA(30*1*1*1; 60*1*1*1; 20*1*1*1)
```

Uma observação importante é que os intervalos de avaliação e o intervalo de soma devem ter a mesma dimensão, ou seja, o mesmo número de elementos.

Funções de Suplemento

CONVERTER ADD

A função CONVERTER_ADD faz a conversão entre unidades de medida. Sua estrutura é similar à função CONVERTER do Microsoft Excel:

=CONVERTER_ADD(VALOR;UNIDADE_ORIGEM;UNIDADE_DESTINO)

Onde:

VALOR, é o valor numérico a ser convertido;

UNIDADE_ORIGEM é o código de representação da unidade de medida atual na qual o VALOR está mensurado;

UNIDADE_DESTINO é o código de representação da unidade de medida para o qual o VALOR será convertido.

Os argumentos UNIDADE_ORIGEM e UNIDADE_DESTINO devem ser preenchidos com códigos da mesma categoria. A lista completa de categorias e suas respectivas unidades é apresentada abaixo.

Peso

Unidade	Representação
Grama	"g"
Libra massa	"sg"
Massa em libras (avoirdupois)	"lbm"
Massa atômica	"u"
Massa em onças (avoirdupois)	"ozm"
Stone	"stone"
Grão	"grain"
Ton (curto)	"ton"
Pennyweight	"pweight"
Hundredweight	"hweight"
Shorthundredweight	"shweight"
Tonelagem de arqueação bruta	"brton"

Hora

Unidade	Representação
Ano	"yr"
Dia	"day"
Hora	"hr"
Minuto	"mn"
Segundo	"sec" ou "s"

Comprimento

Unidade	Representação
Metro	"m"
Milha inglesa	"mi"
Milha náutica	"Nmi"
Polegada	"in"
Pé	"ft"
Jarda	"yd"
Ângstrom	"ang"
Paica	"Pica"
Ano-luz	"ano-luz"
Vara	"EII"
Parsec	"parsec"

Pressão

Unidade	Representação	
Pascal	"Pa" (ou "p")	
Atmosfera	"atm" (ou "at")	
mm de Mercúrio	"mmHg"	
Torricelli	"Torr"	
Libra força por polegada quadrada	"psi"	

Força

Unidade	Representação
Newton	"N"
Dyna	"dyn" (ou "dy")
Libra força	"lbf"
Pond	"pond"

Energia

Unidade	Representação
Joule	"J"
Erg	"e"
Caloria termodinâmica	"c"
Caloria IT	"cal"
Eletron-volt	"eV" (ou "ev")
Cavalo-força-hora	"HPh"
Watt-hora	"Wh" (ou "wh")
Pé-libra	"flb"
BTU	"BTU" (ou "btu")

Potência

Unidade	Representação
Cavalo-força	"HP"
Watt	"W" (ou "w")
Cavalo-vapor	"PS"

Indução magnética

Unidade	Representação
Tesla	"T"
Gauss	"ga"

Temperatura

Unidade	Representação	
Grau Celsius	"C"	
Grau Fahrenheit	"F"	
Kelvin	"K" (ou "kel")	
Réaumur	"Reau"	
Rankine	"Rank"	

Área

Unidade	Representação
Metro quadrado	"m2"
Milha inglesa quadrada	"mi2"
Milha náutica quadrada	"Nmi2"
Polegada quadrada	"in2"
Pé quadrado	"ft2"
Jarda quadrada	"yd2"
Ângstrom quadrado	"ang2"
Paica quadrada	"Pica2"
Acre	"acre"
Hectare	"ha"
Are	"Ar"
Morgen	"Morgen"

Velocidade

Unidade	Representação		
Metros por segundo	"m/s" (ou "m/sec")		
Metros por hora	"m/h"		
Nó	"kn"		
Nó (Grã-Bretanha)	"admkn"		
Milhas por hora	"mph"		

Volume

Unidade	Representação			
Colher de chá	"tsp"			
Colher de sopa	"tbs"			
Onça fluida	"oz"			
Xícara	"cup"			
Pinta estadunidense	"pt" (ou "us_pt")			
Pinta inglesa	"uk_pt"			
Quarto	"qt"			
Galão	"gal"			
Litro	"l" (ou "L" ou "lt")			
Schooner australiano	"Schooner"			
Middy australiano	"Middy"			
Glass australiano	"Glass"			
Metro cúbico	"m3"			
Milha inglesa cúbica	"mi3"			
Milha náutica cúbica	"Nmi3"			
Polegada cúbica	"in3"			
Pé cúbico	"ft3"			
Jarda cúbica	"yd3"			
Ângstrom cúbico	"ang3"			
Paica cúbica	"Pica3"			
Barril estadunidense	"barrel"			
Bushel estadunidense	"bushel"			
Register Ton	"regton"			

Informação

Unidade	Representação		
Bit	"bit"		
Byte	"byte"		

Prefixos

Os seguintes códigos permitem o uso de prefixos para a representação de grandezas: g, u, m, parsec, anoluz, seg, s, Pa, atm,at, mmHg, N, dyn, dy, pond, J, e, c, cal, eV, ev, Wh, wh, W, w, T, ga, K, kel, I, L, It, m3, m2, ang2, ar, m/s, m/sec, bit e byte.

São admitidos prefixos com escala positiva ou negativa, conforme a configuração abaixo.

Escala negativa

Prefixo	у	z	a	f	р	n	u	m	С	d
	yocto	zepto	atto	femto	pico	nano	micro	milli	centi	deci
Escala	10-24	10-21	10-18	10-15	10-12	10-9	10 ⁻⁶	10-3	10-2	10-1

Escala positiva

Prefixo	е	h	k	М	G	Т	Р	E	Z	Υ
	deca	hecto	kilo	mega	giga	tera	peta	exa	zetta	yotta
Escala	10¹	10 ²	10 ³	10 ⁶	10 ⁹	10 ¹²	10 ¹⁵	10 ¹⁸	1021	10 ²⁴

Exemplos

Converter 100 Km para milhas → =CONVERTER_ADD(100;"km";"mi")

Resultado: 62,1371192237

Converter 30°C para °F → =CONVERTER_ADD(30;"C";"F")

Resultado: 86

Converter 234 horas em dias → =CONVERTER_ADD(234;"hr";"day")

Resultado: 9,75

Converter 120 Km/h em Milhas por hora → =CONVERTER_ADD(120;"km/h";"mph")

Resultado: 74,5645430685

SE + É.NÃO.DISP + PROCV

Leia sobre a combinação SE + É.NÃO.DISP + PROCV na página 24, função É.NÃO.DISP.

SOMA e SE em fórmulas matriciais

A combinação das funções SOMA e SE em fórmulas matriciais permite contabilizações que podem considerar vários critérios ao mesmo tempo. Em outras palavras, é como se pudéssemos utilizar as funções SOMASE e CONT.SE com vários critérios, em vez de apenas um.

Os exemplos abaixo são bastante ilustrativos. Considere a seguinte tabela de despesas em um dado mês:

	Α	В	С	D	E
1					
2		Descrição	Conta	Data	Valor
3		Combustível	16496	03/05/12	R\$ 150,00
4		Supermercado	13423	03/05/12	R\$ 15,00
5		Cinema	13423	05/05/12	R\$ 30,00
6		Aluguel	16496	10/05/12	R\$ 500,00
7		Condomínio	13423	10/05/12	R\$ 120,00
8		Energia Elétrica	13423	10/05/12	R\$ 60,00
9		Restaurante	13423	10/05/12	R\$ 62,00
10		Padaria	16496	11/05/12	R\$ 12,00
11		Combustível	16496	16/05/12	R\$ 50,00
12		Restaurante	16496	16/05/12	R\$ 28,00
13		Supermercado	13423	16/05/12	R\$ 25,00
14		Cinema	13423	19/05/12	R\$ 45,00
15		Padaria	16496	19/05/12	R\$ 8,00
16		Supermercado	16496	22/05/12	R\$ 170,00
17		Padaria	16496	22/05/12	R\$ 10,00
18		Restaurante	13423	24/05/12	R\$ 35,00
19		Padaria	13423	29/05/12	R\$ 15,00
20					

Fórmula para contagem

Uma contabilização útil seria, por exemplo, saber quantas operações foram realizadas na conta 13423 a partir do dia 15. Note que, como falamos, a função CONT.SE não resolve o problema pois não permite considerarmos os dois critérios ao mesmo tempo.

A solução, portanto, é utilizarmos a combinação matricial das funções SOMA e SE, tendo a SOMA como função mais externa, seguida de dois SEs concatenados. No SE mais interno, utilizamos 1 para verdadeiro e 0 para falso:


{=SOMA(SE(C3:C19=13423;(SE(DIA(D3:D19)>15;1;0))))}

Essa será uma fórmula matricial. Fórmulas matriciais têm como característica básica o processamento de intervalos de mais de uma célula para a obtenção de um resultado. No nosso exemplo, obteremos, primeiro, todas as contas iguais a 13423. Depois, dentro desse conjunto, todos os dias a partir do dia 15. Para os valores que obedecerem a ambos os critérios, atribuiremos o valor 1. Ao final, somamos esses valores para obtermos o número de 9 operações realizadas na conta 13423 a partir do dia 15.

Para definir a fórmula como matricial, edite-a no **Assistente de funções** e marque a opção **Matricial** no canto inferior esquerdo do diálogo. Outra forma de indicar que a fórmula é matricial é teclar *Ctrl+Shift+Enter* ao final da digitação da fórmula. Ambos os procedimentos adicionarão chaves no início e no final da fórmula.

```
{=SOMA(SE(C3:C19=13423;(SE(DIA(D3:D19)>15;1;0))))}
```

Note que as chaves não são digitadas. A atribuição matricial deve ser feita através de um dos dois métodos descritos acima.



Combinações úteis 59


Fórmula para soma

Imagine, agora, que nosso objetivo seja o de obter a soma de gastos realizados na conta 13423 a partir do dia 15. A fórmula que utilizaremos é similar à anterior. No entanto, agora, colocaremos o intervalo de E3:E19 dentro do parâmetro verdadeiro da função SE mais interna.

```
{=SOMA(SE(C3:C19=13423;(SE(DIA(D3:D19)>15;E3:E19;0))))}
```

Ou seja, faremos a avaliação das contas iguais à 13423 e dos dias maiores que 15. Somente os valores que corresponderem a esses critérios dentro do intervalo E3:E19 serão contabilizados na soma. O resultado será, então, 120.

Essa construção funciona como uma função SOMASE com vários critérios.



Índice alfabético de funções

ABS	27	ÉERROS	24
AGORA	17	ESQUERDA	48
ALEATÓRIO	27	ESTILO	40, 41
ANO	17, 18	EXT.TEXTO	48
ARRED	27, 28	FIMMÊS	18, 19
ARREDONDAR.PARA.BAIXO	27, 28	HOJE	19
ARREDONDAR.PARA.CIMA	28	ÍNDICE	42, 43, 45
BDCONTAR	16	LINHA	42, 44
BDSOMA	15, 16	MAIOR	21
CARACT	46, 49	MAIÚSCULA	48, 49
CONCATENAR	47	MÁXIMO	21, 22, 35
CONT.NÚM	29, 35	MÉDIA	22, 35
CONT.SE	30	MENOR	23
CONT.VALORES	31, 32, 35	MÊS	19
CONTAR.VAZIO	32	MÍNIMO	23, 35
CORRESP	37, 38, 39, 40, 42	NÚM.CARACT	49
DESLOC	39, 40	PRI.MAIÚSCULA	49
DESVPAD	20, 35	PROC	44
DIA	17, 48	PROCV	24, 25, 44, 45
DIASÚTEISTOTAIS	17	SE	24, 25, 26, 30, 41, 51
DIATRABALHOTOTAL	17, 18	SOMARPRODUTO	51, 52
DIREITA	47	SOMASE	16, 34
DOMINGODEPÁSCOA	18	SUBTOTAL	35
É.NÃO.DISP	24, 25, 38	TEXTO	40, 48, 49
ÉANOBISSEXTO	18	TRANSPOR	50
		VALOR	24, 26, 45, 48